ap.

WEATHER BUREAU
Office of Systems Development
Techniques Development Laboratory
Silver Spring, Md.

July 1969

Automatic Decoding of Hourly Weather Reports

Technical Memorandum WBTM TDL 21

U.S. DEPARTMENT OF COMMERCE / ENVIRONMENTAL SCIENCE SERVICES ADMINISTRATION

ESSA TECHNICAL MEMORANDA

Weather Bureau, Techniques Development Laboratory Series

The primary purpose of the Techniques Development Laboratory of the Office of Systems Development is to translate increases in basic knowledge in meteorology and allied disciplines into improved operating techniques and procedures. To achieve this goal, TDL conducts and sponsors applied research and development aimed at improvement of diagnostic and prognostic methods for producing weather information. The laboratory carries out studies both for the general improvement of prediction methodology used in the National Meteorological Service System and for more effective utilization of weather forecasts by the ultimate user.

ESSA Technical Memoranda in the Weather Bureau Techniques Development Laboratory series facilitate rapid distribution of material which may be preliminary in nature and which may be published formally elsewhere at a later date. The first five papers in the TDL series are part of the former Weather Bureau Technical Notes series.

Papers listed below are available from the Clearinghouse for Federal Scientific and Technical Information, U.S. Department of Commerce, Sills Bldg., 5285 Port Royal Road, Springfield, Va. 22151. Price: $3.00 hard copy; $0.65 microfiche. Order by accession number shown in parentheses at end of each entry.

TN 10 TDL 1 Objective Prediction of Daily Surface Temperature. William H. Klein, Curtis W. Crockett, and Carlos R. Dunn, October 1965. (PB-168 590)

TN 11 TDL 2 Hurricane Cindy Galveston Bay Tides. N. A. Pore, A. T. Angelo, and J. G. Taylor, September 1965. (PB-168 608)

TN 29 TDL 3 Atmospheric Effects on Re-Entry Vehicle Dispersions. Karl R. Johannessen, December 1965. (PB-169 381)

TN 45 TDL 4 **A Synoptic Climatology of Winter Precipitation from 700-mb. Lows for the Intermountain Areas of the West. D. L. Jorgensen, W. H. Klein, and A. F. Korte, May 1966. (PB-170 635)**

TN 47 TDL 5 **Hemispheric Specification of Sea Level Pressure from Numerical 700-mb. Height Forecasts. William H. Klein and Billy M. Lewis, June 1966. (PB-173 091)**

WBTM TDL 6 **A Fortran Program for the Calculation of Hourly Values of Astronomical Tide and Time and Height of High and Low Water. N. A. Pore and R. A. Cummings, January 1967. (PB-174 660)**

WBTM TDL 7 **Numerical Experiments Leading to the Design of Optimum Global Meteorological Networks. M. A. Alaka and F. Lewis, February 1967. (PB-174 497)**

WBTM TDL 8 **An Experiment in the Use of the Balance Equation in the Tropics. M. A. Alaka, D. T. Rubsam, and G. E. Fisher, March 1967. (PB-174 501)**

WBTM TDL 9 **A Survey of Studies of Aerological Network Requirements. M. A. Alaka, May 1967. (PB-174 984)**

WBTM TDL 10 **Objective Determination of Sea Level Pressure from Upper Level Heights. William Klein, Frank Lewis, and John Stackpole, May 1967. (PB-179 949)**

WBTM TDL 11 **Short Range, Subsynoptic Surface Weather Prediction. H. R. Glahn and D. A. Lowry, July 1967. (PB-175 772)**

WBTM TDL 12 **Charts Giving Station Precipitation in the Plateau States from 700-Mb. Lows During Winter. D. L. Jorgensen, A. F. Korte, and J. A. Bunce, Jr., October 1967. (PB-176 742)**

WBTM TDL 13 **Interim Report on Sea and Swell Forecasting. N. A. Pore and W. S. Richardson, December 1967. (PB-177 038)**

WBTM TDL 14 Meteorological Analysis of 1964-65 ICAO Turbulence Data. DeVer Colson, September 1968. (PB-180 268)

WBTM TDL 15 Prediction of Temperature and Dew Point by Three-Dimensional Trajectories. Ronald M. Reap, September 1968. (PB-180 727)

(Continued on inside back cover)

U.S. DEPARTMENT OF COMMERCE
Environmental Science Services Administration
Weather Bureau

ESSA Technical Memorandum WBTM TDL 21

AUTOMATIC DECODING OF HOURLY WEATHER REPORTS

George W. Hollenbaugh, Harry R. Glahn, Dale A. Lowry

OFFICE OF SYSTEMS DEVELOPMENT
TECHNIQUES DEVELOPMENT LABORATORY

SILVER SPRING, MD.
July 1969

UDC 551.509.1:681.3.06
551.5 Meteorology
.509 Synoptic analysis
.1 Meteorological codes
681.3 Automatic data processing
.3.06 Computer programs

CONTENTS

AUTOMATIC DECODING OF HOURLY WEATHER REPORTS

George W. Hollenbaugh, Harry R. Glahn, Dale A. Lowry

ABSTRACT

A computer program which performs automatic decoding of weather reports is described. The program was designed to be run operationally on the ESSA CDC 6600 computer and includes error detection. Hourly airways data are collected on magnetic tape via the IBM 360-40 computer. Details of the search procedure and decoding of the data are described. A listing of the program is included.

INTRODUCTION

This computer program decodes hourly weather reports, and furnishes the data to the Subsynoptic Advection Model (SAM) [2] developed by the Techniques Development Laboratory. The reports are collected from the operational weather communication circuits by an IBM 360-40 computer at the National Meteorological Center (NMC), and put on magnetic tape. The decoding program is written in the FORTRAN language and is run operationally on the ESSA CDC 6600 computer. The design of the program follows, in many respects, that described by Marx and Shroyer [3] of the Travelers Research Center. Figure 1 shows a sample hourly aviation report and offers an explanation of the contents.

SUBROUTINE DESCRIPTIONS

The decoding program consists of a family of subroutines. It was designed to handle the primary sequence reports contained in the 15 Service "A" circuits 8021 thru 8035 as well as relayed reports from Air Force, Navy, and Canadian operational circuits. Reports are transmitted sequentially each hour H on the hour. The scan for the airways collection (SA) begins promptly at H+00 minutes and all reports are received centrally by H+27 minutes. From approximately H+27 to H+35 minutes the Air Force, Navy, and Canadian reports are received.

HAR S -X 10⊕M 20⊕120⊕7/8R+F 042/53/51/1804/994/F3

Station call letters	Harrisburg, Pa.
Report type	Record special
Obscuration based on earth's surface	Partly obscured
Lowest cloud layer and height of base	Scattered cloud layer based 1000 feet above the earth's surface
Ceiling classification	Measured Ceiling
Ceiling cloud layer and height of base	Broken cloud layer based 2000 feet above the earth's surface
Highest cloud layer and height of base	Overcast cloud layer based 12000 feet above the earth's surface
Visibility (horizontal)	7/8 of a mile
Weather and obstructions to vision	Heavy rain and fog
Sea-level pressure	1004.2 millibars
Temperature	53 degrees F.
Dew point	51 degrees F.
Wind direction	From 180 degrees (south)
Wind speed	4 knots
Altimeter setting	29.94 inches of mercury
Remarks	Sky 3/10 obscured by fog

Figure 1. A sample hourly aviation report.

The routines in the decoding program are:

1. Control program (DECODE)
2. Data reading routine (RDADIS)
3. Identification function (BOSS)
4. Date-time identification (SHAD)
5. Transfer routine (TRNSFR)
6. Station identification (DENTFY)
7. Special routine (SPECAL)
8. Wind identification (WINDS)
9. Dew point and temperature routines (DEWPT and TEMPR)
10. Sea-level pressure identification (PRESS)
11. Weather identification and remarks (WEATHR and REMARK)
12. Visibility identification (VISBTY)
13. Sky cover identification (SKYCVR)
14. Ceiling height identification (CELING)

Each of the routines will be discussed in general terms in the following sections; the FORTRAN listings are given in the appendix. The order in which they are discussed is not necessarily the order of execution.

1. Control Program (DECODE)

The control program ties all of the sub-programs together and is referred to as the calling program. This program manipulates the entry of each sub-program. It is also the function of the control program to call the data reading routine (RDADIS) which starts the flow of raw data into the computer. Each of the sub-programs can begin to perform only on orders from the calling program (except RDADIS) and each returns control to the calling program when its individual functions have been completed. Some of the sub-programs return a condition word to the calling program which indicates the type of problem, if any, it encountered with the data. Then the calling program determines which routine to enter depending on the condition word.

2. Data Reading Routine (RDADIS)

This subroutine controls the flow of data into the computer. As data are needed, the subroutine CAK01 (which is available with the SCOPE Operating System used by the ESSA Computer Division) is called to read the BCD data from magnetic tape. The tape record size is 200 BCD characters; the first 192 characters are data and the last 8 are filler. This routine is called by BOSS, SHAD, and TRNSFR, as well as DECODE.

3. Identification Function (BOSS)

Meteorological observations are transmitted over the communication lines in large blocks. At the beginning of each block is usually a collection heading. Each collection heading consists of a specific beginning of message condition code, data heading "SA", and the date-time group. Following the collection heading are the hourly aviation weather reports of all stations in the area circuit data block. Each station also has a specific beginning of message condition code. The purpose of this subroutine is to examine the condition code so as to identify the type of information available in the message.

4. Date-Time Identification (SHAD)

The object of the routine is to verify the identification "SA" along with its date-time group. The verifying of the date-time group is done by decoding it from the heading and comparing it to a specific date-time group supplied to the routine.

5. Transfer Routine (TRNSFR)

The impulses that are received at a collection point from the weather communication circuit are a combination of printing and non-printing characters. Many of the non-printing characters are used only for communication management and present somewhat of a problem. The solution to the problem is to eliminate as many of these superfluous characters as possible without destroying those characters which aid the computer program to recognize and classify the significant information. The routine goes through a character search on each hourly station report and removes such characters as letter and figure shifts, extra spaces, end of line codes, etc. which serve no useful purpose to the program. The edited report is then placed in a work area. If, during the process of placing the report in the work area, there is found to be more than 200 characters transferred before an end of message select code is found, the message is rejected as not being an airways report. Also, if the end of message select code is found and only 10 or less characters have been transferred, the message is rejected.

6. Station Identification (DENTFY)

When the program finds a possible hourly report, the station identification routine makes a search for a unique set of call letters. In some cases, there may be two sets of call letters perhaps separated by an end of line condition code. This indicates that the station whose call letters appear first has transmitted an hourly report for that station whose call letters appear second. In this case, the first set of call letters is rejected and the second set retained as being that of the reporting station. These call letters are then compared with a list (directory) of possible station call letters for rejection or acceptance of the report.

7. Special Routine (SPECAL)

The system is designed specifically to handle hourly aviation reports. Therefore, it does not have the capability of processing all types of aviation messages transmitted over the communication circuits. It is the function of the SPECAL sub-program to recognize and accept only "record" or "record special" observations. This is done by examining for a time group following the station call letters and preceding the first cloud layer or obscuration symbol.

8. Wind Identification (WINDS)

Whether an observation is decoded or not is dependent on whether the wind group can be identified. The routine first proceeds to locate this key group which is composed of four digits preceded and followed by slashes (/). This format is unique inasmuch as no other variable of the report can have this arrangement of characters. Once this variable is located, the wind direction is checked to see that it is between 0 and 360 degrees. Also, if the direction is zero, the speed must be zero and vice versa. If any one of the tests fails, the entire report is rejected.

9. Dew Point and Temperature Routines (DEWPT and TEMPR)

The purpose of these two routines is to search the report and identify the dew point (DEWPT) and temperature (TEMPR). The search procedure is from right to left in the work area beginning at the wind. The dew point and temperature are found in that order by first examining for a slash followed by one, two, or three numerical characters possibly preceded by a minus sign. Although the dew point routine will accept a three digit number, a value is considered to be correct only if it is in the range $-50 \leq DP \leq 90$. The temperature acceptance range is $-40 \leq TP \leq 110$. Both temperature and dew point are indicated as missing (9999) if either one is missing or out of range or if the temperature is lower than the dew point.

10. Sea-Level Pressure Identification (PRESS)

The purpose of this routine is to identify the sea-level pressure. The search procedure is from right to left in the work area beginning at the temperature. However, if either the dew point or temperature routine had trouble locating its variable, the search procedure is from left to right beginning after the station call letters. The sea-level pressure is identified by a space followed by three digits and a slash. This variable is not checked for errors except through the objective analysis program [1] that uses it.

11. Weather Identification and Remarks (WEATHR and REMARK)

This routine identifies and decodes the surface weather portion of the report by testing the alphabetical characters, if any, between the last cloud or obscuration symbol and the next following space. The weather is put into a simple code: 1 = frozen precipitation, 2 = drizzle or freezing drizzle, 3 = rain, freezing rain, or thunder and 0 = none of the above. If no weather is found in the body of the report, the remarks are searched by the routine REMARK for precipitation having begun and ended, having ended, or occasional precipitaion having occurred during the hour. Any weather noted in the remarks is coded in the same manner as mentioned above. If multiple precipitation types are observed, the one commanding the highest code figure is used. For example, when mixed rain and snow occurs, code 3 is used rather than code 1.

12. Visibility Identification (VISBTY)

The purpose of this routine is to identify and check the visibility. The search procedure is to test the characters following the last cloud or obscuration symbol, up to the weather or next space. A check of this variable is that whenever the prevailing visibility is less than 7 miles, a weather and/or obstruction to vision group must be included in the report.

13. Sky Cover Identification (SKYCVR)

This routine identifies the last cloud layer or obscuration symbol, which represents the total sky cover. The search is made from right to left. If no symbol is found, a missing indicator (9999) is stored by the routine for the variable. The total cloud cover is put into the code: 0 = clear or partial obscuration, 1 = scattered, 2 = broken, 3 = overcast, and 4 = obscured.

14. Ceiling Height Identification (CELING)

This routine identifies the ceiling height. The lowest cloud layer not based on the surface which equals or exceeds 6/10 opaque coverage is considered to constitute a ceiling at the cloud base. Obscuring phenomena (10/10 coverage) are considered to be surface based and constitute a ceiling determined by the vertical visibility. Ceiling heights must be preceded by an indication as to how the height of the ceiling was determined (E for estimated, M for measured, etc.). The cloud symbol for amount and the reported height indicator are tested for correctness. Also, the height is rejected if there are more than three digits in the group or if it is greater than 600. A cirriform cloud ceiling with height unknown is indicated as 7777. An unlimited ceiling condition is indicated as 8888.

DATA INPUT IN COMMON

Upon entry to the subroutine DECODE, certain information must be present. The following list specifies the names of the variables and array elements to which values must be assigned in COMMON.

DAT	Date-time group of the hourly reports in the BCD form 2R00 + Hour * 4096 + day * 16777216 where hour is in GMT.
NOSTA	Number of stations in station directory, maximum of 600.
STALST (600)	BCD call letters (right justified) of station directory.

DATA OUTPUT IN COMMON

The output data are in a form and format which is easily used by an analysis [1] or prediction routine. When certain elements are missing or in error from a particular observation, the number 9999 is stored by the program for that variable, except for pressure when a zero is stored. It is therefore possible for the analysis program to have information available from observing stations regardless of whether the particular observation contained valid data for all groups. A listing of the array elements and an explanation of each is presented below:

HDATA (600, 1)	Sea-level pressure in millibars
HDATA (600, 2)	Dew point temperature in degrees Fahrenheit
HDATA (600, 3)	Weather in code form
HDATA (600, 4)	Total sky cover in code form
HDATA (600, 5)	Surface temperature in degrees Fahrenheit
HDATA (600, 6)	Surface wind direction in degrees
HDATA (600, 7)	Surface wind speed in knots
HDATA (600, 8)	Height of the ceiling in hundreds of feet
HDATA (600, 9)	Visibility in statute miles.

COMMENTS

It was found during the early stages of checkout that a problem existed in connection with identifying the "broken" sky cover symbols. The external BCD bit configuration on the magnetic tape for this symbol is 16 octal. Unfortunately, there is no internal "Display Code" bit configuration (under the present SCOPE System being used for the ESSA CDC 6600) which corresponds to 16 octal. Therefore, this 6-bit group was being converted by default to 00 octal which is not a legal BCD character in the SCOPE System. In order that the DECODE statement could be used, these 00 octal groups were changed to 77 octal. Consequently, 77 octal is recognized by the SKYCVR subroutine as a broken sky cover symbol.

If, during the decoding of many stations, a report is found which has previously been decoded, the previous information is replaced by the newly decoded information.

Some automatic reporting stations have different reporting formats. Therefore, difficulty may be encountered in decoding their reports.

REFERENCES

1. Glahn, H. R., G. W. Hollenbaugh, and D. A. Lowry, "An Operationally Oriented Objective Analysis Program," ESSA Technical Memorandum WBTM TDL 22, July 1969, 20 pp.

2. Glahn, H. R., D. A. Lowry, and G. W. Hollenbaugh, "An Operational Subsynoptic Advection Model," ESSA Technical Memorandum WBTM TDL 23, July 1969, 26 pp.

3. Marx, Henry P., and Robert M. Shroyer, "The Development of Computer Input Data Handling (IDH) Systems for the Processing of Hourly and Other Operational Weather Data," Technical Memorandum No. 6, Contract No. FAA BRD-363, The Travelers Research Center, July 1961, 36 pp.

APPENDIX

This appendix contains the listings of sixteen subroutines in FORTRAN language necessary for performing the decoding of hourly weather reports. These subroutines are in the form for use with the SCOPE Operating System on the CDC 6600 at Suitland, Maryland.

Subroutine DECODE is called when hourly weather report decoding is desired and it, in turn, calls the other fifteen routines as necessary.

```
      SUBROUTINE DECODE
C     JULY, 1968        HOLLENBAUGH
C     TO EFFECT THE EDITING, DECODING, AND ERROR CHECKING OF SERVICE
C         A HOURLY WEATHER REPORTS THROUGH THE USE OF SUBROUTINES
C     TAPE USE, FORTRAN NUMBER
C         (7) = HOURLY REPORTS (ADIS READOUT BY 360-40 COMPUTER)
C     DIMENSIONED VARIABLES AND COMMON BLOCKS
C                 IS=POSITION OR COUNT IN EDITED WEATHER MESSAGE AT
C                    END OF WIND GROUP BEGINNING FROM THE RIGHT
C                 IT=SUBSCRIPT LOCATION OF STATION CALL LETTERS IN
C                    DIRECTORY WITH RESPECT TO REPORTING STATION
C                 KK=NUMBER OF CHARACTERS EXAMINED IN EDITED WEATHER
C                    MESSAGE BEGINNING FROM THE RIGHT
C                 JJ=NUMBER OF CHARACTERS EXAMINED IN EDITED WEATHER
C                    MESSAGE BEGINNING FROM THE LEFT
C                 NN=NUMBER OF CHARACTERS EXAMINED IN UNEDITED
C                    WEATHER MESSAGE BEGINNING FROM THE LEFT
C               DAT=DATE-TIME GROUP OF HOURLY REPORTS (INPUT)
C              DATE=DECODED DATE-TIME GROUP FROM COLLECTION HEADING
C              STCL=CONTAINS THE CALL LETTERS OF REPORTING STATION
C              ICND=CONDITION WORD, RELATIVE TO EACH ROUTINE
C            NOSTA=NUMBER OF STATIONS IN STATION DIRECTORY
C            NCHRT=NUMBER OF CHARACTERS IN EDITED WEATHER MESSAGE
C            BF(  )=200 CHARACTER RECORD PUT INTO 20 FULL WORDS
C            WK(  )=WORK AREA OF EDITED WEATHER REPORT
C            CH(  )=1 CHARACTER/WORD, DECODED FROM 20 WORD RECORD
C         MTAG(  )=(1)-STATION DATA DECODED, (0)-REPORT MISSING
C         STAY(  )=STATION Y-POSITION IN TERMS OF GRID
C         STAX(  )=STATION X-POSITION IN TERMS OF GRID
C       STALST(  )=BCD CALL LETTERS (RIGHT JUSTIFIED), DIRECTORY
C       HDATA( ,1)=STATION VALUES OF PRESSURE (MILLIBARS AND TENTHS)
C       HDATA( ,2)=STATION VALUES OF DEW POINT (WHOLE DEGS F)
C       HDATA( ,3)=STATION VALUES OF CODED PRIORITY WEATHER
C            (0.)-FOR NO WEATHER OR NO SIGNIFICANT WEATHER PRESENT
C            (1.)-FOR ANY OF THE FOLLOWING(AP,A,IC,SG,SP,SW,S,EW,E)
C            (2.)-FOR ANY OF THE FOLLOWING(ZL,L)
C            (3.)-FOR ANY OF THE FOLLOWING(ZR,RW,R,T)
C       HDATA( ,4)=STATION VALUES OF CODED TOTAL SKY COVER
C            (0.)-FOR CLEAR OR PARTIAL OBSCURATION CONDITION
C            (1.)-FOR SCATTERED CONDITION
C            (2.)-FOR BROKEN CONDITION
C            (3.)-FOR OVERCAST CONDITION
C            (4.)-FOR OBSCURATION CONDITION
C       HDATA( ,5)=STATION VALUES OF TEMPERATURE (WHOLE DEGS F)
C       HDATA( ,6)=STATION VALUES OF WIND DIRECTION (DEGS)
C       HDATA( ,7)=STATION VALUES OF WIND SPEED (KNOTS)
C       HDATA( ,8)=STATION VALUES OF CEILING HEIGHT (HUNDREDS OF FT)
C            (7777.)-CIRRIFORM CLOUDS, HEIGHT UNKNOWN
C            (8888.)-UNLIMITED CEILING CONDITION
C       HDATA( ,9)=STATION VALUES OF VISIBILITY (MILES)
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),WK(200),BF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C     ********************************************************
      ICND=MK=0
   99 MI=MJ=ML=0
      CALL CAK08(7,KS)
      CALL CAK03(7,KS,2)
      IF(KS)100,110,100
  100 PRINT 105
  105 FORMAT(* ADIS TAPE SKIPPING PROBLEM, PROCESSING CONTINUED*)
  110 NN=245
      CALL RDADIS
      IF(ICND)120,120,150
  120 NN=49
  125 CALL BOSS
      IF(MK)127,127,128
  127 GO TO(140,125,135),ICND
  128 GO TO(300,175,125),ICND
  135 CALL SHAD
      GO TO(140,160,165,170),ICND
  140 IF(MJ.EQ.1.OR.ML.EQ.1)145,150
  145 PRINT 146
  146 FORMAT(10/,*  EOF FOUND, NO ADIS DATA DECODED.*)
      RETURN
```

```
150   PRINT 151
151   FORMAT(10/,*  EOF FOUND, NO ADIS DATA ON TAPE.*)
      RETURN
160   PRINT 161
161   FORMAT(*0DATE-TIME GROUP COULD NOT BE DECODED, NON-NORMAL FORMA
      MJ=1
      GO TO 125
165   MK=1
      IF(MJ.EQ.1.OR.ML.EQ.1)166,125
166   PRINT 167
167   FORMAT(*0DATE-TIME GROUPS CORRECT, ADIS REWOUND, ALL DATA USED.
      GO TO 99
170   PRINT 171,DATE,DAT
171   FORMAT(*0DATE-TIME GROUPS INCORRECT.  ADIS=*,R6,* OTHER=*,R6)
      ML=1
      GO TO 125
175   CALL TRNSFR
      IF(ICND-2)300,125,180
180   CALL DENTFY
      IF(ICND-2)125,125,183
183   CALL SPECAL
      IF(ICND-2)125,125,185
185   CALL WINDS(IDIR,ISPD)
      IF(ICND-2)125,125,190
190   MTAG(IT)=1
      HDATA(IT,6)=IDIR
      HDATA(IT,7)=ISPD
      CALL DEWPT
      IF(ICND-2)215,200,195
195   CALL TEMPR
      IF(ICND-2)215,200,200
200   CALL PRESS
      IF(ICND-2)220,205,205
205   CALL WEATHR
      IF(ICND-2)225,207,207
207   CALL VISBTY
      IF(ICND-2)230,208,208
208   CALL SKYCVR
      IF (ICND-2)235,209,209
209   IF(HDATA(IT,4)-1.)210,210,211
210   HDATA(IT,8)=8888.
      GO TO 212
211   CALL CEILNG
      IF(ICND-2)240,212,212
212   IF(HDATA(IT,3))213,213,125
213   CALL REMARK
      GO TO 125
215   MTAG(IT)=0
      HDATA(IT,6)=9999.
      HDATA(IT,7)=9999.
      HDATA(IT,2)=9999.
      HDATA(IT,5)=9999.
220   HDATA(IT,1)=0.
225   HDATA(IT,3)=9999.
230   HDATA(IT,9)=9999.
235   HDATA(IT,4)=9999.
240   HDATA(IT,8)=9999.
      GO TO 125
300   RETURN
      END
```

000135 CARDS

```
      SUBROUTINE RDADIS
C        JULY, 1968        HOLLENBAUGH
C        READS AND DECODES THE HOURLY WEATHER REPORTS
C        CONDITION WORD
C          0=NEW DATA READ AND DECODED
C          1=END OF FILE DETECTED, ALL DATA READ
      DIMENSION A(10),L(10),IF(20),IH(240)
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),WK(200),BF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
      EQUIVALENCE (BF,IF),(CH,IH),(A,L)
      DATA(A(J),J=1,10)/77000000000000000000B,00770000000000000000B,
     100007700000000000000B,00000077000000000000B,00000000770000000000B,
     100000000007700000000B,00000000007700000000B,00000000000077000000B,
     100000000000000077000B,00000000000000007700B,00000000000000000077B/
C        ********************************************************
  100 ICND=0
      M=NN-5
      DO 115 I=M,240
      IH(I-192)=IH(I)
  115 CONTINUE
      CALL CAKQ1(7,KS,BF,20,0,0)
      IF(KS)120,150,120
  120 IKS=KS.AND.30B
      IF(IKS)125,130,125
  125 ICND=1
      RETURN
  130 IKS=KS.AND.300000B
      IF(IKS)135,140,135
  135 PRINT 136
  136 FORMAT(* ADIS TAPE LENGTH ERROR, PROCESSING CONTINUED*)
      GO TO 150
  140 IKS=KS.AND.4000B
      IF(IKS)145,150,145
  145 PRINT 146
  146 FORMAT(* ADIS TAPE PARITY ERROR, PROCESSING CONTINUED*)
  150 DO 160 I=1,20
      DO 155 J=1,10
      JKL=IF(I).AND.L(J)
      IF(JKL)155,153,155
  153 IF(I)=IF(I).OR.L(J)
  155 CONTINUE
  160 CONTINUE
      DECODE(100,175,BF)(CH(I),I=49,148)
      DECODE(92,180,BF(11))(CH(I),I=149,240)
  175 FORMAT(100R1)
  180 FORMAT(92R1)
      NN=NN-192
      RETURN
      END
```

000050 CARDS

```
      SUBROUTINE BOSS
C        JULY, 1968        HOLLENBAUGH
C        LOCATES BEGINNING OF MESSAGE 'CONDITION' CODE
C        CONDITION WORD
C              1=END OF FILE DETECTED BY RDADIS ROUTINE
C              2=BEGINNING OF STATION MESSAGE FOUND
C              3=BEGINNING OF COLLECTION MESSAGE FOUND
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),IH(240),WK(200),PF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C          ************************************************************
      ICND=0
  100 N=NN
  105 IF(N+10-240)110,110,160
C          BEGINNING OF COLLECTION MESSAGE SEQUENCE
  110 IF(IH(N).EQ.1R*.AND.IH(N+1).EQ.1R*.AND.
     1 IH(N+2).EQ.1R(.AND.IH(N+3).EQ.1R))120,130
  120 NN=N+4
      ICND=3
      RETURN
C          BEGINNING OF STATION MESSAGE SEQUENCE
  130 IF(IH(N).EQ.1R*.AND.IH(N+1).EQ.1R*.AND.IH(N+2).EQ.1R()140,150
  140 NN=N+3
      ICND=2
      RETURN
  150 N=N+1
      GO TO 105
  160 NN=N
      CALL RDADIS
      IF(ICND)100,100,170
  170 RETURN
      END
```

000032 CARDS

```
      SUBROUTINE SHAD
C        NOVEMBER, 1968  HOLLENBAUGH
C        DECODES AND INTERPRETS THE COLLECTION HEADING
C        CONDITION WORD
C              1=END OF FILE DETECTED BY RDADIS ROUTINE
C              2=DATE-TIME GROUP COULD NOT BE DECODED (NON-NORMAL FORMAT)
C              3=DATE-TIME GROUP FOUND AND IS CORRECT
C              4=DATE-TIME GROUP FOUND BUT IS INCORRECT
      DIMENSION JK(12)
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
     1 NCHRT,NOSTA,ICND,STCL,IATE,IAT,NN,JJ,KK,IT,IS
C          ***************************************************
      ICND=0
  100 N=NN
      IF(N+30-240)110,175,175
  110 IF(IH(N).EQ.1RS.AND.IH(N+1).EQ.1RA)135,130
  130 N=N+1
      IF(N-NN-10)110,170,170
  135 N=N+2
      J=N+11
      L=0
      DO 145 I=N,J
      IF(IH(I).GE.1R0.AND.IH(I).LE.1R9)140,145
  140 L=L+1
      JK(L)=IH(I)
  145 CONTINUE
      N=N+L+1
      IF(L.GE.6.AND.L.LE.8)155,170
  155 IATE=JK(L)+JK(L-1)*64+JK(L-2)*4096+JK(L-3)*262144+
     1 JK(L-4)*16777216+JK(L-5)*1073741824
      IF(IAT-IATE)190,180,190
  170 ICND=2
      NN=N
      RETURN
  175 NN=N
      CALL RDADIS
      IF(ICND)100,100,185
  180 ICND=3
      NN=N
  185 RETURN
  190 ICND=4
      NN=N
      RETURN
      END
```

000045 CARDS

```
      SUBROUTINE TRNSFR
C         JULY, 1968        HOLLENBAUGH
C         EDITS THE HOURLY WEATHER REPORT
C         CONDITION WORD
C            1 = END OF FILE DETECTED BY RDADIS ROUTINE
C            2 = WEATHER REPORT COULD NOT BE EDITED
C            3 = WEATHER REPORT EXAMINED AND EDITED
C         MARK=1 - A SPACE HAS PREVIOUSLY BEEN TRANSFERRED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),IH(240),IK(200),BE(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C            ********************************************************
  100 J=1
      MARK=0
      ICND=0
  103 N=NN
  105 IF(N+10-240)110,200,200
C         END OF MESSAGE SEQUENCE
  110 IF(IH(N)-1R$)125,115,125
  115 N=N+1
      IF(IH(N)-1R*)105,120,105
  120 N=N+1
      IF(IH(N)-1R()133,203,133
C         BEGINNING OF MESSAGE SEQUENCE
  125 IF(IH(N)-1R*)150,130,150
  130 N=N+1
  133 IF(IH(N)-1R*)140,135,140
  135 N=N+1
      IF(IH(N)-1R()140,210,140
C         END OF LINE SEQUENCE
  140 IF(IH(N)-1R$)105,142,105
  142 IF(J-7)144,144,170
  144 NN=N+1
      GO TO 100
C         SPECIAL CHARACTERS
  150 IF(IH(N)-1R()155,170,155
  155 IF(IH(N)-1R))160,170,160
  160 IF(IH(N)-1R )180,165,180
  165 IF(MARK)170,185,170
  170 N=N+1
      GO TO 105
  180 MARK=0
      GO TO 190
  185 MARK=1
  190 IK(J)=IH(N)
      J=J+1
      N=N+1
      IF(J-200)105,195,195
  195 ICND=2
      NN=N
      RETURN
  200 NN=N
      CALL RDADIS
      IF(ICND)103,103,220
  203 N=N+1
      GO TO 215
  210 N=N-2
  215 ICND=3
      NN=N
  220 NCHRT=J-1
      RETURN
      END
```

000062 CARDS

```
      SUBROUTINE DENTFY
C        JULY, 1968       HOLLENBAUGH
C        LOCATE AND DECODE STATION CALL LETTERS OF REPORTING STATION
C        CONDITION WORD
C              1 = NO STATION CALL LETTERS
C              2 = STATION CALL LETTERS NOT IN DIRECTORY
C              3 = STATION CALL LETTERS IN DIRECTORY
      DIMENSION IL(5)
      COMMON DATUM,MTALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),RE(20),
     1 NCHRT,NOSTA,ICND,ITCL,DATE,DAT,NN,JJ,KK,IT,IS
C     ********************************************************
      ICND=0
      IF(NCHRT-4)100,100,105
  100 ICND=1
      RETURN
  105 L=0
      J=1
      IF(IK(J)-1R )115,110,115
  110 J=J+1
  115 M=J+3
      DO 140 I=J,M
      IF(IK(I).GE.1RA.AND.IK(I).LE.1RZ)135,130
  130 IF(IK(I)-1R )100,135,100
  135 L=L+1
      IL(L)=IK(I)
      IF(IK(I)-1R )140,145,140
  140 CONTINUE
      GO TO 100
  145 IF(L.GE.3.AND.L.LE.4)150,100
  150 JJ=I
      ITCL=IL(3)+IL(2)*64+IL(1)*4096
      IF(ITCL.EQ.3RGLS.OR.ITCL.EQ.3RORH)100,155
C        THE ABOVE STATEMENT IS TO ELIMINATE PICKING UP
C        DATA FOR TWO AMOS STATIONS BECAUSE OF DIFFICULTY WITH
C        THEIR REPORTING FORMAT
  155 DO 210 I=1,NOSTA
      IF(MTALST(I)-ITCL)210,225,210
  210 CONTINUE
      ICND=2
      RETURN
  225 IT=I
      ICND=3
      RETURN
      END
```

000045 CARDS

```
      SUBROUTINE SPECAL
C        JULY, 1968        HOLLENBAUGH
C        IDENTIFIES A 'RECORD' OR 'RECORD SPECIAL' OBSERVATION
C        CONDITION WORD
C              1 = NO STATION INFORMATION PRESENT
C              2 = A REPORT OTHER THAN A 'RECORD' OR 'RECORD SPECIAL'
C              3 = AN ACCEPTABLE STATION REPORT FOUND
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(20,3,3)
      COMMON/BLOCK6/MTAC(600),HDATA(600,9),CH(240),IK(200),BF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C             ***************************************************
  100 ICND=0
      IF(JJ+6-NCHRT)110,105,105
  105 ICND=1
      RETURN
  110 J=JJ
      M=JJ
  115 IF(IK(J).EQ.1RS.AND.IK(J+1).GT.1RZ)117,120
  117 M=J+1
      GO TO 130
  120 IF(IK(J).EQ.54B.OR.IK(J).EQ.64B)132,125
  125 IF(IK(J).EQ.74B.OR.IK(J).EQ.77B.OR.IK(J).EQ.30B)132,130
  130 J=J+1
      IF(NCHRT/2-J)105,105,115
  132 IF(IK(J+1).GE.1R0.AND.IK(J+1).LE.1R9)135,130
  135 NUMB=0
      DO 150 L=M,J
      IF(IK(L).GE.1R0.AND.IK(L).LE.1R9)140,145
  140 NUMB=NUMB+1
      IF(NUMB-4)150,155,155
  145 NUMB=0
  150 CONTINUE
      GO TO 160
  155 IF(M-JJ)105,160,175
  160 IF(IK(J)-1R )165,170,165
  165 J=J-1
      IF(J-JJ)105,170,160
  170 JJ=J
      ICND=3
      RETURN
  175 ICND=2
      RETURN
      END
```

000043 CARDS

```
      SUBROUTINE WINDS(IDIR,ISPD)
C        NOVEMBER, 1968  HOLLENBAUGH
C        LOCATE AND DECODE THE WIND GROUP
C        CONDITION WORD
C              1 = NO WIND GROUP COULD BE FOUND
C              2 = SUSPECTED WIND GROUP FOUND BUT NON-NORMAL FORMAT
C              3 = WIND GROUP FOUND AND DECODED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),BE(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C          ***************************************************
  100 ICND=0
      NUMB=0
      DO 135 K=JJ,NCHRT
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)120,130
  120 NUMB=NUMB+1
      IF(NUMB-4)135,150,150
  130 NUMB=0
  135 CONTINUE
      ICND=1
      RETURN
  150 IS=K+1
      K=K-4
      IF(IK(K)-1R/)152,160,152
  152 IF(IK(K).EQ.1R .AND.IK(K-1).EQ.1R/)160,180
  160 IF(IS-NCHRT)165,165,170
  165 IF(IK(IS).GE.1R0.AND.IK(IS).LE.1R9)180,170
  170 IDIR=(IK(K+1)-33B)*100+(IK(K+2)-33B)*10
      ISPD=(IK(K+3)-33B)*10+(IK(K+4)-33B)
      IF(IDIR-360)195,190,180
  180 ICND=2
      KK=K
      RETURN
  190 IF(ISPD)180,180,205
  195 IF(IDIR)180,200,190
  200 IF(ISPD)180,205,180
  205 ICND=3
      KK=K
      RETURN
      END
```

000040 CARDS

```
      SUBROUTINE DEWPT
C        JULY, 1968        HOLLENBAUGH
C        LOCATE AND DECODE THE DEW POINT
C        CONDITION WORD
C              1 = MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C              2 = DEW POINT CANNOT BE FOUND OR DECODED
C              3 = DEW POINT FOUND AND DECODED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),RF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C     **************************************************
  100 ICND=0
      K=KK
      IF(K-JJ)115,115,125
  115 ICND=1
      RETURN
  125 IF(IK(K)-1R/)130,160,130
  130 IF(IK(K)-1R )145,135,145
  135 K=K-1
      IF(IK(K)-1R/)145,160,145
  145 ICND=2
      HDATA(IT,2)=9999.
      HDATA(IT,5)=9999.
      RETURN
  160 K=K-1
      IF(IK(K)-1R )170,165,170
  165 K=K-1
  170 IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)175,145
  175 IDPU=IK(K)-33B
      K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)185,255
  185 IDPT=IK(K)-33B
      K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)200,235
  200 IDPH=IK(K)-33B
      DP=IDPH*100+IDPT*10+IDPU
  215 K=K-1
  220 ICND=3
      HDATA(IT,2)=DP
      KK=K
      RETURN
  235 IF(IK(K)-1R-)270,240,270
  240 DP=-IDPT*10-IDPU
      GO TO 215
  255 IF(IK(K)-1R-)280,260,280
  260 DP=-IDPU
      GO TO 215
  270 DP=IDPT*10+IDPU
      GO TO 220
  280 DP=IDPU
      GO TO 220
      END
```

000052 CARDS

```
      SUBROUTINE TEMPR
C        JULY, 1968       HOLLENBAUGH
C        LOCATE AND DECODE THE TEMPERATURE
C        CONDITION WORD
C             1 = MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C             2 = TEMPERATURE CANNOT BE FOUND OR DECODED, OR THE
C                 DEW POINT IS INCORRECT
C             3 = TEMPERATURE FOUND AND DECODED, AND VERIFIES DEW POINT
C                 IS CORRECT
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),BF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C          ************************************************
  100 ICND=0
      K=KK
      IF(K-JJ)115,115,125
  115 ICND=1
      RETURN
  125 IF(IK(K)-1R/)130,145,130
  130 IF(IK(K)-1R )280,135,280
  135 K=K-1
      IF(IK(K)-1R/)280,145,280
  145 K=K-1
  150 IF(IK(K)-1R )165,155,165
  155 K=K-1
  165 IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)170,280
  170 ITPU=IK(K)-33B
      K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)180,245
  180 ITPT=IK(K)-33B
      K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)195,220
  195 ITPH=IK(K)-33B
      TP=ITPH*100+ITPT*10+ITPU
  210 K=K-1
      GO TO 265
  220 IF(IK(K)-1R-)235,225,235
  225 TP=-ITPT*10-ITPU
      GO TO 210
  235 TP=ITPT*10+ITPU
      GO TO 265
  245 IF(IK(K)-1R-)260,250,260
  250 TP=-ITPU
      GO TO 210
  260 TP=ITPU
  265 IF(TP-HDATA(IT,2))280,275,270
  270 IF(TP-HDATA(IT,2)-50.)275,275,280
  275 JTP=TP
      JDP=HDATA(IT,2)
      IF(JTP.LT.-40.OR.JTP.GT.110.OR.JDP.LT.-50.OR.JDP.GT.90)280,295
  280 ICND=2
      HDATA(IT,2)=9999.
      HDATA(IT,5)=9999.
      RETURN
  295 ICND=3
      HDATA(IT,5)=TP
      KK=K
      RETURN
      END
```

000059 CARDS

```
      SUBROUTINE PRESS
C        JULY, 1968        HOLLENBAUGH
C        LOCATE AND DECODE THE PRESSURE GROUP
C        CONDITION WORD
C              1 = MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C              2 = PRESSURE GROUP CANNOT BE FOUND OR DECODED
C              3 = PRESSURE GROUP FOUND AND DECODED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),RF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C     ***************************************************
  100 K=KK
      IF(K-JJ)110,110,120
  110 ICND=1
      RETURN
  120 IF(ICND-2)110,265,125
C        BEGIN EXAMINING FOR PRESSURE (NORMAL MESSAGE)
  125 IF(IK(K)-1R/)130,141,130
  130 IF(IK(K)-1R )240,135,240
  135 K=K-1
      IF(IK(K)-1R/)240,141,240
  141 K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)150,240
  150 IPRU=IK(K)-33B
      K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)160,240
  160 IPRT=IK(K)-33B
      K=K-1
      IF(IK(K).GE.1R0.AND.IK(K).LE.1R9)180,240
  180 IPRH=IK(K)-33B
      K=K-1
      IF(IK(K).EQ.1R .OR.IK(K).EQ.1RF)205,210
  205 K=K-1
  210 P=IPRH*100+IPRT*10+IPRU
      IF(P-500.)230,220,220
  220 HDATA(IT,1)=900.+P/10.
      GO TO 255
  230 HDATA(IT,1)=1000.+P/10.
      GO TO 255
  240 HDATA(IT,1)=0.
      ICND=2
      RETURN
  255 ICND=3
      KK=K
      RETURN
C        BEGIN EXAMINING FOR PRESSURE (NON-NORMAL MESSAGE)
  265 L=K
  270 IF(IK(L).EQ.54B.OR.IK(L).EQ.64B)285,275
  275 IF(IK(L).EQ.74B.OR.IK(L).EQ.77B.OR.IK(L).EQ.30B)285,280
  280 L=L-1
      IF(L-JJ)240,270,270
  285 NUMB=0
      DO 300 I=L,K
      IF(IK(I).GE.1R0.AND.IK(I).LE.1R9)290,295
  290 NUMB=NUMB+1
      IF(NUMB-3)300,305,305
  295 NUMB=0
  300 CONTINUE
      GO TO 240
  305 K=I+1
      IF(IK(K)-1R/)310,141,310
  310 IF(IK(K).EQ.1R .AND.IK(K+1).EQ.1R/)141,240
      END
```

000063 CARDS

```
      SUBROUTINE WEATHR
C        JULY, 1968        HOLLENBAUGH
C        LOCATE AND DECODE THE LEADING WEATHER SYMBOL
C        CONDITION WORD
C              1 = MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C              2 = NO WEATHER OR NO SIGNIFICANT WEATHER FOUND
C              3 = SIGNIFICANT WEATHER FOUND AND DECODED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C             ************************************************
  100 ICND=0
      K=KK
  110 IF(K-JJ)115,125,125
  115 ICND=1
      RETURN
  125 IF(IK(K).EQ.54B.OR.IK(K).EQ.64B)140,126
  126 IF(IK(K).EQ.74B.OR.IK(K).EQ.77B)140,127
  127 IF(IK(K)-1RX)130,128,130
  128 IF(IK(K-1).GE.1R0.AND.IK(K-1).LE.1R9.OR.IK(K-1).EQ.1R-)140,115
  130 K=K-1
      GO TO 110
  140 J=K
      IF(J-KK)150,155,115
  150 K=K+1
C        BEGIN TESTING FOR SPECIAL CHARACTERS
  155 IF(IK(K).EQ.1R .OR.IK(K).EQ.1RU.OR.IK(K).EQ.1RV)160,158
  158 IF(IK(K)-1RB)190,170,190
  160 K=K+1
      IF(K-KK)155,155,170
  170 HDATA(IT,3)=0.
      ICND=2
      KK=J
      RETURN
C        BEGIN TESTING FOR WEATHER GROUP 1
  190 IF(IK(K).EQ.1RA.OR.IK(K).EQ.1RI)210,200
  200 IF(IK(K).EQ.1RS.OR.IK(K).EQ.1RE)210,250
  210 HDATA(IT,3)=1.
      ICND=3
      KK=J
      RETURN
C        BEGIN TESTING FOR WEATHER GROUP 2
  250 IF(IK(K)-1RL)280,255,280
  255 HDATA(IT,3)=2.
      ICND=3
      KK=J
      RETURN
C        BEGIN TESTING FOR WEATHER GROUP 3
  280 IF(IK(K).EQ.1RR.OR.IK(K).EQ.1RT.OR.IK(K).EQ.1RW)295,160
  295 HDATA(IT,3)=3.
      ICND=3
      KK=J
      RETURN
      END
```

000054 CARDS

```
      SUBROUTINE REMARK
C        JULY, 1968       HOLLENBAUGH
C        LOCATE AND DECODE THE REMARKS PERTAINING TO PRECIP-
C             ITATION BEGINNING AND/OR ENDED, OR OCCASIONAL PRECIP
C        CONDITION WORD
C             1 = MESSAGE BUFFER EXHAUSTED, ON ENTERING ROUTINE
C             2 = NO PRECIPITATION FOUND
C             3 = PRECIPITATION FOUND AND DECODED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),RF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C          ************************************************
  100 ICND=0
      IF(IS+2-NCHRT)110,110,105
  105 ICND=1
      RETURN
  110 I=IS
  115 IF(IK(I)-1RE)120,130,120
  120 IF(IK(I)-1RL)125,155,125
  125 I=I+1
      IF(I-NCHRT)115,270,270
C        TEST FOR PRECIPITATION ENDED
  130 IF(IK(I+1).GE.1R0.AND.IK(I+1).LE.1R9)133,125
  133 IF(IK(I+2).GE.1R0.AND.IK(I+2).LE.1R9)135,125
  135 IF(IK(I-1).GE.1R0.AND.IK(I-1).LE.1R9)145,140
  140 J=I-1
      GO TO 180
C        TEST FOR PRECIPITATION BEGINNING AND ENDED
  145 IF(IK(I-2).GE.1R0.AND.IK(I-2).LE.1R9.AND.IK(I-3).EQ.1RB)150,125
  150 J=I-4
      GO TO 180
C        TEST FOR OCNL PRECIPITATION
  155 IF(IK(I-1).EQ.1RN.AND.IK(I-2).EQ.1RC)160,125
  160 IF(IK(I-3).EQ.1R0.AND.IK(I+1).EQ.1R )165,125
  165 IF(IK(I+2)-1RZ)170,175,170
  170 J=I+2
      GO TO 190
  175 J=I+3
      GO TO 190
C        TEST FOR WEATHER GROUP 1
  180 IF(IK(J).EQ.1RA.OR.IK(J).EQ.1RI)205,185
  185 IF(IK(J).EQ.1RS.OR.IK(J).EQ.1RE)205,215
  190 IF(IK(J).EQ.1RS.OR.IK(J).EQ.1RE)195,220
  195 IF(IK(J+1).EQ.1RW.OR.IK(J+1).EQ.1R+)205,200
  200 IF(IK(J+1).EQ.1R-.OR.IK(J+1).EQ.1R )205,125
  205 IF(HDATA(IT,3)-1.)210,125,125
  210 HDATA(IT,3)=1.
      GO TO 125
C        TEST FOR WEATHER GROUP 2
  215 IF(IK(J)-1RL)240,230,240
  220 IF(IK(J)-1RL)245,225,245
  225 IF(IK(J+1).EQ.1R+.OR.IK(J+1).EQ.1R-.OR.IK(J+1).EQ.1R )230,125
  230 IF(HDATA(IT,3)-2.)235,125,125
  235 HDATA(IT,3)=2.
      GO TO 125
C        TEST FOR WEATHER GROUP 3
  240 IF(IK(J).EQ.1RT.OR.IK(J).EQ.1RR)260,125
  245 IF(IK(J)-1RR)125,250,125
  250 IF(IK(J+1).EQ.1RW.OR.IK(J+1).EQ.1R+)260,255
  255 IF(IK(J+1).EQ.1R-.OR.IK(J+1).EQ.1R )260,125
  260 HDATA(IT,3)=3.
  265 ICND=3
      RETURN
  270 IF(HDATA(IT,3))275,275,265
  275 ICND=2
      RETURN
      END
```

000067 CARDS

```
      SUBROUTINE VISBTY
C        JULY, 1968       HOLLENBAUGH
C        LOCATE AND DECODE THE VISIBILITY
C        CONDITION WORD
C           1=MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C           2=VISIBILITY CANNOT BE FOUND OR DECODED
C           3=VISIBILITY FOUND AND DECODED
      DIMENSION IFIG(5)
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,2)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),RE(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C          ***************************************************
  100 ICND=0
      K=KK
      IF(K-JJ)110,115,115
  110 ICND=1
      RETURN
  115 L=0
      M=1
      J=K+1
      N=K+5
      DO 135 I=J,N
      IF(IK(I)-1R/)125,123,125
  123 M=M+L
      GO TO 135
  125 IF(IK(I).GE.1R0.AND.IK(I).LE.1R9)130,140
  130 L=L+1
      IFIG(L)=IK(I)-33B
  135 CONTINUE
  138 HDATA(IT,9)=9999.
      ICND=2
      RETURN
  140 GO TO (145,160,175,138),M
  145 IF(L-2)150,155,138
  150 HDATA(IT,9)=IFIG(L)
      GO TO 180
  155 HDATA(IT,9)=IFIG(L-1)*10+IFIG(L)
      GO TO 180
  160 IF(L-2)138,165,170
  165 HDATA(IT,9)=FLOATF(IFIG(L-1))/FLOATF(IFIG(L))
      GO TO 180
  170 HDATA(IT,9)=FLOATF(IFIG(L-2))/FLOATF(IFIG(L-1)*10+IFIG(L))
      GO TO 180
  175 IF(L-3)138,176,138
  176 HDATA(IT,9)=FLOATF(IFIG(L-2))+FLOATF(IFIG(L-1))/FLOATF(IFIG(L)
  180 IF(HDATA(IT,9)-7.)185,200,200
  185 IF(IK(I).EQ.IR .OR.IK(I).EQ.IRU.OR.IK(I).EQ.IRV)190,195
  190 I=I+1
  195 IF(IK(I).LE.IRZ.AND.IK(I).NE.IRM)200,138
  200 ICND=3
      RETURN
      END
```

000052 CARDS

```
      SUBROUTINE SKYCVR
C        JULY, 1968       HOLLENBAUGH
C        LOCATE AND DECODE THE TOTAL SKY COVER OR OBSCURATION SYMBOL
C        CONDITION WORD
C              1 = MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C              2 = SKY COVER SYMBOL CANNOT BE FOUND OR DECODED
C              3 = SKY COVER SYMBOL FOUND AND DECODED
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),BE(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C        ************************************************************
  100 ICND=0
      K=KK
      IF(K-JJ)120,130,130
  120 ICND=1
      RETURN
C        BEGIN TESTING FOR A CLEAR SYMBOL
  130 IF(IK(K)-54B)145,135,145
  135 HDATA(IT,4)=0.
      GO TO 225
C        BEGIN TESTING FOR A SCATTERED SYMBOL
  145 IF(IK(K)-64B)160,147,160
  147 IF(IK(K+1)-1RU)150,180,150
  150 HDATA(IT,4)=1.
      GO TO 225
C        BEGIN TESTING FOR A OVERCAST SYMBOL
  160 IF(IK(K)-74B)175,165,175
  165 HDATA(IT,4)=3.
      GO TO 225
C        BEGIN TESTING FOR A BROKEN SYMBOL
  175 IF(IK(K)-77B)195,180,195
  180 HDATA(IT,4)=2.
      GO TO 225
C        BEGIN TESTING FOR AN OBSCURATION
  195 IF(IK(K)-1RX)240,200,240
  200 IF(K-1-JJ)240,210,210
  210 IF(IK(K-1)-1R-)215,135,215
  215 IF(IK(K-1).GE.1R0.AND.IK(K-1).LE.1R9)220,240
  220 HDATA(IT,4)=4.
  225 ICND=3
      KK=K
      RETURN
  240 K=K-1
      IF(K-JJ)250,130,130
  250 HDATA(IT,4)=9999.
      ICND=2
      KK=K
      RETURN
      END
```

000049 CARDS

```
      SUBROUTINE CFLING
C        JULY, 1968       HOLLENBAUGH
C        LOCADE AND DECODE THE CLOUD CEILING HEIGHT
C        CONDITION WORD
C              1 = MESSAGE BUFFER EXHAUSTED ON ENTERING ROUTINE
C              2 = CEILING HEIGHT CANNOT BE FOUND OR DECODED
C              3 = CEILING HEIGHT FOUND AND DECODED
      DIMENSION JK(4)
      COMMON DATUM,STALST(600),STAX(600),STAY(600),JTAB1(38,3,3)
      COMMON/BLOCK6/MTAG(600),HDATA(600,9),CH(240),IK(200),BF(20),
     1 NCHRT,NOSTA,ICND,STCL,DATE,DAT,NN,JJ,KK,IT,IS
C          ************************************************
  100 ICND=0
      IF(KK-JJ)105,110,110
  105 ICND=1
      RETURN
  110 DO 130 L=JJ,KK
      IF(IK(L).EQ.1RV.OR.IK(L).EQ.1RX)130,125
  125 IF(IK(L).GE.1RA.AND.IK(L).LE.1RZ)135,130
  130 CONTINUE
      HDATA(IT,8)=8888.
      ICND=3
      RETURN
  135 IF(IK(KK).EQ.54B.OR.IK(KK).EQ.64B)145,140
  140 IF(IK(KK).EQ.74B.OR.IK(KK).EQ.77B.OR.IK(KK).EQ.30B)145,143
  143 HDATA(IT,8)=9999.
      ICND=2
      RETURN
  145 IF(IK(L).EQ.1RM.OR.IK(L).EQ.1RA.OR.IK(L).EQ.1RR.OR.IK(L).EQ.1RB.
     1 OR.IK(L).EQ.1RW.OR.IK(L).EQ.1RD.OR.IK(L).EQ.1RE.OR.IK(L).EQ.1RP)
     2 160,150
  150 IF(IK(L).EQ.1RU.AND.IK(L+1).EQ.54B.OR.IK(L+1).EQ.64B)156,155
  155 IF(IK(L).EQ.1RU.AND.IK(L+1).EQ.74B.OR.IK(L+1).EQ.77B)156,143
  156 HDATA(IT,8)=7777.
      ICND=3
      RETURN
  160 IF(IK(L+1).GE.1R0.AND.IK(L+1).LE.1R9)161,143
  161 J=0
      M=L+1
      DO 170 I=M,KK
      IF(IK(I).GE.1R0.AND.IK(I).LE.1R9)165,175
  165 J=J+1
      JK(J)=IK(I)-33B
      IF(J-3)170,170,143
  170 CONTINUE
  175 IF(IK(I).EQ.54B.OR.IK(I).EQ.64B)180,177
  177 IF(IK(I).EQ.74B.OR.IK(I).EQ.77B.OR.IK(I).EQ.30B)180,178
  178 IF(IK(I)-1RV)143,179,143
  179 I=I+1
      GO TO 175
  180 GO TO (185,190,195),J
  185 HDATA(IT,8)=JK(J)
      GO TO 200
  190 HDATA(IT,8)=JK(J-1)*10+JK(J)
      GO TO 200
  195 HDATA(IT,8)=JK(J-2)*100+JK(J-1)*10+JK(J)
  200 IF(HDATA(IT,8)-600.)205,205,143
  205 ICND=3
      RETURN
      END
```

000060 CARDS

(Continued from inside front cover)

WBTM TDL 16 Objective Visibility Forecasting Techniques Based on Surface and Tower Observations. Donald M. Gales, October 1968. (PB-180 479)

WBTM TDL 17 Second Interim Report on Sea and Swell Forecasting. N. A. Pore and Lt. W. S. Richardson, USESSA, January 1969. (PB-182 273)

WBTM TDL 18 Conditional Probabilities of Precipitation Amounts in the Conterminous United States. Donald L. Jorgensen, William H. Klein, and Charles F. Roberts, March 1969. (PB-183 144)

WBTM TDL 19 An Operationally Oriented Small-Scale 500-Millibar Height Analysis. Harry R. Glahn, March 1969.

WBTM TDL 20 A Comparison of Two Methods of Reducing Truncation Error. Robert J. Bermowitz, May 1969. (PB-184 741)


DEPARTMENT OF COMMERCE
UNITED STATES OF AMERICA